# Maintainer: Lukas Fleischer <lfleischer@archlinux.org>
# Contributor: Aaron Bull Schaefer <aaron@elasticdog.com>
# Contributor: Felix Hanley <felix@seconddrawer.com.au>

pkgname=spawn-fcgi
pkgver=1.6.4
pkgrel=4
pkgdesc='A FastCGI process manager for web applications'
arch=('loong64' 'x86_64')
url='https://redmine.lighttpd.net/projects/spawn-fcgi/'
license=('BSD')
depends=('glibc')
source=("https://www.lighttpd.net/download/${pkgname}-${pkgver}.tar.bz2"{,.asc})
md5sums=('bf269849995ab8b3a8989779fa5818aa'
         'SKIP')
validpgpkeys=(C7CA1E9E29DC77F5480894B2E0E7D0171E95BAD7) # Stefan Bühler <stefan@stbuehler.de>

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  make DESTDIR="${pkgdir}" install
  install -Dm0644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
